NAVAL POSTGRADUATE SCHOOL
Monterey, California

THESIS

ORIENTATION GUIDANCE AND CONTROL FOR MARINE VEHICLES IN THE HORIZONTAL PLANE

by

Prouttichai Suwandee

June, 1991

Thesis Advisor: Fotis A. Papoulias

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE

REPORT DOCUMENTATION PAGE

Form Approved OMB No 0704 0188

1a REPORT SECURITY CLASSIFICATION: Unclassified	1b RESTRICTIVE MARKINGS
2a SECURITY CLASSIFICATION AUTHORITY	3 DISTRIBUTION/AVAILABILITY OF REPORT: Approved for public release distribution is unlimited
2b DECLASSIFICATION/DOWNGRADING SCHEDULE	
4 PERFORMING ORGANIZATION REPORT NUMBER(S)	5 MONITORING ORGANIZATION REPORT NUMBER(S)
6a NAME OF PERFORMING ORGANIZATION: Naval Postgraduate School	6b OFFICE SYMBOL (If applicable)
7a NAME OF MONITORING ORGANIZATION: Naval Postgraduate School	
6c ADDRESS (City, State, and ZIP Code)	7b ADDRESS (City, State, and ZIP Code)
8a NAME OF FUNDING/SPONSORING ORGANIZATION	8b OFFICE SYMBOL (If applicable)
9 PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER	
8c ADDRESS (City, State, and ZIP Code)	10 SOURCE OF FUNDING NUMBERS

PROGRAM ELEMENT NO	PROJECT NO	TASK NO	WORK UNIT ACCESSION NO

11 TITLE (Include Security Classification)

ORIENTATION GUIDANCE AND CONTROL FOR MARINE VEHICLES IN THE HORIZONTAL PLANE

12 PERSONAL AUTHOR(S)

Prouttichai Suwandee

13a TYPE OF REPORT	13b TIME COVERED	14 DATE OF REPORT (Year, Month, Day)	15 PAGE COUNT
M.S. Thesis	FROM ____ TO ____	June 1991	49

16 SUPPLEMENTARY NOTATION

17 COSATI CODES			18 SUBJECT TERMS (Continue on reverse if necessary and identify by block number)
FIELD	GROUP	SUB-GROUP	Autonomous vehicles, Guidance and control, Stability, Path keeping

19 ABSTRACT (Continue on reverse if necessary and identify by block number)

A pure pursuit guidance law and a heading autopilot are coupled in order to provide path control of submersibles or surface ships in the horizontal plane. Proper design of the combined scheme allows for accurate path keeping during straight line motion. The simulation results are extended to cover cases of step changes in the desired path. The scheme provides a viable alternative to cross track error autopilots.

20 DISTRIBUTION/AVAILABILITY OF ABSTRACT: ☒ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT ☐ DTIC USERS	21 ABSTRACT SECURITY CLASSIFICATION: Unclassified	
22a NAME OF RESPONSIBLE INDIVIDUAL: Fotis A. Papoulias	22b TELEPHONE (Include Area Code): (408) 6463381	22c OFFICE SYMBOL: ME/PA

DD Form 1473, JUN 86 Previous editions are obsolete SECURITY CLASSIFICATION OF THIS PAGE

S/N 0102-LF-014-6603 Unclassified

Orientation Guidance and Control for Marine Vehicles in the Horizontal Plane

by

Prouttichai Suwandee
Lieutenant Commander, Royal Thai Navy
B.S., Royal Thai Navy Academy, Thailand

Submitted in partial fulfillment
of the requirements for the degree of

MASTER OF SCIENCE IN MECHANICAL ENGINEERING

from the

NAVAL POSTGRADUATE SCHOOL
June 1991

Author: ______________________
Prouttichai Suwandee

Approved by: ______________________
Fotis A. Papoulias, Thesis Advisor

____
Anthony J. Healey, Chairman
Department of Mechanical Engineering

ABSTRACT

A pure pursuit guidance law and a heading autopilot are coupled in order to provide path control of submersibles or surface ships in the horizontal plane. Proper design of the combined scheme allows for accurate path keeping during straight line motion. The simulation results are extended to cover cases of step changes in the desired path. The scheme provides a viable alternative to cross track error autopilots.

TABLE OF CONTENTS

ACKNOWLEDGEMENT

The continuous support and patient guidance from my advisor, professor Fotis A. Papoulias motivated me very much to work on this research area. His unfailing attention and concern contributed significantly to the success of my research.

I. INTRODUCTION

In any operational scenario of an underwater vehicle there exits a triple-nested sequence of mission accomplishment operations: Path planning, navigation, guidance, and autopilot design. The path planner takes information from charted obstacles and friendly or hostile environments and generates a smooth plan for the vehicle to follow. A certain level of feedback exists in this operation through the use of sonar beams in order to replan a path when uncharted objects are encountered or when the mission requirements have changed. Based on the desired vehicle positions and orientations at certain points, several classes of smooth paths containing sets of straight line segments, and circular arcs or cubic splines can be obtained [1]. Once a smooth path is generated, the navigator provides through a selected guidance law the appropriate vehicle heading commands which are in turn delivered by the autopilot. Line of sight guidance using a discrete series of way points was studied by Lienard in [2] using sliding mode heaving, propulsion, and depth keeping autopilots. The scheme demonstrated excellent stability and robustness characteristics, although the actual vehicle response was found to lag significantly the commanded straight line paths. The guidance and autopilot functions can be combined when the lateral deviation off the desired path is directly incorporated into the control law design. This leads to the development of a cross track error autopilot. Such schemes have been studied by Chism [3] and Hawkinson [4] for the single input and multiple input case, respectively. Cross track error autopilots provide more accurate path keeping response but they must be designed more carefully since they tend to be more dependent than heading controllers on an accurate description of the vehicle hydrodynamic characteristics. The main reason for this is the increase in the system dimensionality by one. Underwater vehicles operate in changing environments over a wide range of operating speeds and, therefore, a certain degree of uncertainty exists in the vehicle dynamic modeling. Cross track error autopilots also require accurate positional information updates at the same rate as heading and heading rate.

For these reasons, in this work we go back to the case of a heading autopilot coupled with an orientation guidance law. The two main tasks on which we will concentrate are as follows: First, we must develop a way of establishing stability of the combined autopilot/guidance scheme for straight line commanded motions. Second, the actual vehicle response characteristics must be made to resemble the desired cross track error response with smooth transitions between consecutive straight line segment and with minimal path overshoot. A linear full state feedback control law is used to adjust the heading of the vehicle to any desired value, and a pure pursuit guidance law is used to provide the commanded heading for straight line motion. In this scheme the vehicle commanded heading equals the line of sight angle between the vehicle center and a target point moving on the desired path at a constant lookahead distance from the vehicle. This parallels the case studied in [5] except that in our case the existence of lateral and rotational dynamics add more complications to the problem. All computations are performed for the Swimmer Delivery Vehicle [6] for which a set of hydrodynamic characteristics and geometric properties is available. Problem formulation and equations of motion are presented in Section II. The analysis procedure is outlined in Section III, and simulation results are presented in the Section IV. Finally, conclusions and recommendations for further research are given in Section V.

II. DEVELOPMENT OF THE MATHEMATICAL MODEL

A. EQUATIONS OF MOTION

In a moving coordinate frame fixed at the vehicle center, Newton's equations of motion for a rigid body in the horizontal plane are

$$m(\dot{v} + ur + x_G\dot{r} - y_G r^2) \quad = \quad Y \ , \tag{2.1}$$

$$I_Z\dot{r} + mx_G(\dot{v} + ur) - my_G vr \quad = \quad N \ , \tag{2.2}$$

where

v = sway(lateral) velocity,

r = yaw(angular) velocity,

u = forward(surge) speed,

Y = sway force,

N = yaw moment,

m = vehicle mass,

I_Z = vehicle mass moment of inertia,

(x_G, y_G) = coordinates of center of gravity.

Expanding the force Y and moment N in added mass, damping, and drag terms, equations (2.1) and (2.2) are written as

$$m(\dot{v} + ur + x_G \dot{r} - y_g r^2) = \frac{\rho}{2} l^4 Y_{\dot{r}} \dot{r} + \frac{\rho}{2} l^3 (Y_{\dot{v}} \dot{v} + Y_r ur)$$

$$+ \frac{\rho}{2} l^2 Y_v uv - \frac{\rho}{2} \int C_{D_y} h(x) \frac{(v + xr)^3}{| v + xr |} dx + \frac{\rho}{2} l^2 y_\delta u^2 \delta, \tag{2.3}$$

$$I_Z \dot{r} + m x_G (\dot{v} + ur) - m y_G vr = \frac{\rho}{2} l^5 N_{\dot{r}} \dot{r} + \frac{\rho}{2} l^3 (N_{\dot{v}} \dot{v} + N_r ur)$$

$$+ \frac{\rho}{2} l^3 N_v uv - \frac{\rho}{2} \int C_{D_y} h(x) \frac{(v + xr)^3}{| v + xr |} x dx + \frac{\rho}{2} l^3 y_\delta u^2 \delta, \tag{2.4}$$

where

ρ = water density,

l = vehicle length,

δ = rudder angle,

h(x) = vehicle height distribution,

C_{Dy} = drag coefficient.

The inertial position of the vehicle (x,y) and its heading angle ψ (see Figure 1) are given by

$$\dot{\psi} = r , \tag{2.5}$$

$$\dot{x} = u \cos \psi - v \sin \psi , \tag{2.6}$$

$$\dot{y} = u \sin \psi + v \cos \psi . \tag{2.7}$$



Figure 1 Top view of the vehicle

B. STATE SPACE EQUATIONS

Choosing (ψ, v, r) as the state vector, the linearized state space equations (2.3), (2.4), and (2.5) are written as

$$\dot{\psi} = r \ , \tag{2.8}$$

$$\dot{v} = a_{11}uv + a_{12}ur + b_1u^2\delta \ , \tag{2.9}$$

$$\dot{r} = a_{21}uv + a_{22}ur + b_2u^2\delta \ . \tag{2.10}$$

The coefficients in equations (8), (9), and (10) are given by

$$a_{11} = \frac{(I_z - 0.5\rho l^5 N_{\dot{r}})(0.5\rho l^2 y_v) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(0.5\rho l^3 N_v)}{(I_z - 0.5\rho l^5 N_{\dot{r}})(m - 0.5\rho l^3 y_{\dot{v}}) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(mX_G - 0.5\rho l^4 N_{\dot{v}})}$$

$$a_{12} = \frac{(I_z - 0.5\rho l^5 N_{\dot{r}})(-m + 0.5\rho l^3 y_r) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(-mX_G + 0.5\rho l^4 N_r)}{(I_z - 0.5\rho l^5 N_{\dot{r}})(m - 0.5\rho l^3 y_{\dot{v}}) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(mX_G - 0.5\rho l^4 N_{\dot{v}})}$$

$$a_{21} = \frac{(mX_G - 0.5\rho l^4 y_{\dot{r}})(0.5\rho l^3 N_v) - (mX_G - 0.5\rho l^4 N_{\dot{v}})(-m + 0.5\rho l^2 y_v)}{(I_z - 0.5\rho l^5 N_{\dot{r}})(m - 0.5\rho l^3 y_{\dot{v}}) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(mX_G - 0.5\rho l^4 N_{\dot{v}})}$$

$$a_{22} = \frac{(mX_G - 0.5\rho l^4 y_{\dot{r}})(-mX_G + 0.5\rho l^4 N_r) - (mX_G - 0.5\rho l^4 N_{\dot{v}})(-m + 0.5\rho l^3 y_r)}{(I_z - 0.5\rho l^5 N_{\dot{r}})(m - 0.5\rho l^3 y_{\dot{v}}) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(mX_G - 0.5\rho l^4 N_{\dot{v}})}$$

$$b_1 = \frac{(I_z - 0.5\rho l^5 N_{\dot{r}})(0.5\rho l^2 y_\delta) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(0.5\rho l^3 N_\delta)}{(I_z - 0.5\rho l^5 N_{\dot{r}})(m - 0.5\rho l^3 y_{\dot{v}}) - (mX_G - 0.5\rho l^4 y_{\dot{r}})(mX_G - 0.5\rho l^4 N_{\dot{v}})}$$

$$b_2 = \frac{(mX_G - 0.5\rho l^4 y_{\dot r})(0.5\rho l^3 N_\delta) - (mX_G - 0.5\rho l^4 N_{\dot v})(0.5\rho l^2 y_\delta)}{(I_z - 0.5\rho l^5 N_{\dot r})(m - 0.5\rho l^3 y_{\dot v}) - (mX_G - 0.5\rho l^4 y_{\dot r})(mX_G - 0.5\rho l^4 N_{\dot v})}$$

where

$y_\delta,\ y_r,\ y_v,\ y_{\dot r},\ y_{\dot v}$ = *lateral hydrodynamic coefficients*

$N_\delta,\ N_r,\ N_v,\ N_{\dot r},\ N_{\dot v}$ = *yaw hydrodynamic coefficients*

Equations (2.8), (2.9), and (2.10) describe the dynamics of the system with respect to small deviations around a nominal direction $\psi = 0$.

C. PATH KEEPING DEVELOPMENT

1. Heading control

A linear full state feedback control law is of the form

$$\delta = k_1\psi + k_2 v + k_3 r \tag{2.11}$$

where k_1, k_2 and k_3 are the three gains.

From equation (2.8),(2.9),(2.10) and (2.11), the closed loop system is

$$\dot\psi = r \tag{2.12}$$

$$\dot v = b_1 u^2 k_1 \psi + (a_{11} u + b_1 u^2 k_2) v + (a_{12} u + b_1 u^2 k_3) r \tag{2.13}$$

$$\dot{r} = b_2u^2k_1\psi+(a_{21}u+b_2u^2k_2)v+(a_{22}u+b_2u^2k_3)r \tag{2.14}$$

The characteristic equation of (2.12), (2.13) and (2.14) is

$$\begin{vmatrix} 0-\lambda & 0 & 1 \\ b_1u^2k_1 & a_{11}u+b_1u^2k_2-\lambda & a_{12}u+b_1u^2k_3 \\ b_2u^2k_2 & a_{21}u+b_2u^2k_2 & a_{22}u+b_2u^2k_3-\lambda \end{vmatrix} = 0$$

$$\begin{aligned} &\lambda[(a_{11}u+b_1u^2k_2-\lambda)(a_{22}u+b_2u^2k_3-\lambda)-(a_{12}u+b_1u^2k_3)(a_{21}u+b_2u^2k_2)] \\ &\quad -b_1u^2k_1(a_{21}u+b_2u^2k_2)+b_2u^2k_1(a_{11}u+b_1u^2k_2-\lambda) = 0 \end{aligned}$$

$$\begin{aligned} &\lambda[a_{11}a_{22}u^2+a_{11}b_2u^3k_3-\lambda a_{11}u+b_1a_{22}u^3k_2+b_1b_2u^4k_2k_3-\lambda b_1u^2k_2-\lambda a_{22}u \\ &\quad -\lambda b_2u^2k_3+\lambda^2-a_{12}a_{21}u^2-a_{12}b_2u^3k_2-a_{21}b_1u^3k_3-b_1b_2u^4k_2k_3] \\ &\quad -b_1a_{21}u^3k_1-b_1b_2u^4k_1k_2+a_{11}b_2u^3k_1+b_1b_2u^4k_1k_2-\lambda b_2u^2k_1 = 0 \end{aligned}$$

$$\begin{aligned} &\lambda^3-(a_{11}u+b_1u^2k_2+a_{22}u+b_2u^2k_3)\lambda^2+(a_{11}a_{22}u^2+a_{11}b_2u^3k_3+b_1a_{22}u^3k_2 \\ &\quad -a_{12}a_{21}u^2-a_{12}b_2u^3k_2-a_{21}b_1u^3k_3-b_2u^2k_1)\lambda+b_2a_{11}u^3k_1-b_1a_{21}u^3k_1 = 0 \end{aligned} \tag{2.15}$$

2. Desired characteristic equation

The 3rd order ITAE polynomial is defined by

$$\lambda^3+1.75\omega_0\lambda^2+2.15\omega_0^2\lambda+\omega_0^3 = 0 \tag{2.16}$$

$$\omega_0 = 10\frac{ul}{t_c}$$

where

t_c = settling time (dimensionless)

From equations (2.15) and (12.6) we get

$$-a_{11}u-b_1u^2k_2-a_{22}u-b_2u^2k_3 = 1.75\omega_0 \tag{2.17}$$

$$\begin{aligned}&a_{11}a_{22}u^2+a_{11}b_2u^3k_3+b_1a_{22}u^3k_2-a_{12}a_{21}u^2\\&-a_{12}b_2u^3k_2-a_{21}b_1u^3k_3-b_2u^2k_1 = 2.15\omega_0^2\end{aligned} \tag{2.18}$$

$$b_2a_{11}u^3k_1-b_1a_{21}u^3k_1 = \omega_0^3 \tag{2.19}$$

The system of equations (2.17), (2.18) and (2.19) can be solved for the three gains

$$k_1 = \frac{\omega_0^3}{b_2a_{11}u^3-b_1a_{21}u^3}$$

$$k_2 = \frac{u^4(-1.75\omega_0-a_{11}-a_{22})(a_{11}b_2-a_{21}b_2)-b_2u^2(2.15\omega_0^2-a-11a_{22}u^2+a_{12}a_{21}u^2+b_2u^2k_1)}{b_1u^5(a_{11}b_2-a_{21}b_1)-b_2u^5(a_{22}b_1-a_{12}b_2)}$$

$$k_3 = \frac{b_2u^2[2.15\omega_0^2-u^2(a_{11}a_{22}-a_{12}a_{21})+b_2u^2k_1]-u^3(-1.75\omega_0^2-a_{11}u-a_{22}u)(a_{22}b_1-a_{12}b_2)}{b_1u^5(a_{11}b_2-a_{21}b_1)-b_2u^5(a_{22}b_1-a_{12}b_2)}$$

3. Pure pursuit navigation

For a pursuit navigation

$$\psi_c = \sigma \tag{2.20}$$

where

ψ_c = commanded heading

Referring to Figure 1, the line of sight angle σ is defined by

$$\sigma = -\tan^{-1}\frac{y}{x_d} \tag{2.21}$$

where x_d is the vehicle lookahead distance, and the control law (2.11) becomes

$$\delta = k_1(\psi-\psi_c)+k_2 v+k_3 r \tag{2.22}$$

Using equations (2.20), (2.21) and (2.22) we get

$$\delta = k_1(\psi+\tan^{-1}\frac{y}{x_d})+k_2 v+k_3 r \tag{2.23}$$

The linearized equation (2.23) becomes

$$\delta = k_1(\psi+\frac{y}{x_d})+k_2 v+k_3 r \tag{2.24}$$

The linearized equation for the lateral deviation y is obtained from (2.7) as

$$\dot{y} = u\psi+v \tag{2.25}$$

Now the complete state vector is ψ, v, r and y , and the state space equations are written as

$$\dot{\psi} = r \tag{2.26}$$

$$\dot{v} = b_1 u^2 k_1 \psi + (a_{11} u + b_1 u^2 k_2) v + (a_{12} u + b_1 u^2 k_3) r + b_1 u^2 k_1 \frac{1}{x_d} y \tag{2.27}$$

$$\dot{r} = b_2 u^2 k_1 \psi + (a_{21} u + b_2 u^2 k_2) v + (a_{22} u + b_2 u^2 k_3) r + b_2 u^2 k_1 \frac{1}{x_d} y \tag{2.28}$$

$$\dot{y} = u\psi + v \tag{2.29}$$

and the characteristic equation is

$$\begin{vmatrix} 0-\lambda & 0 & 1 & 0 \\ b_1 u^2 k_1 & a_{11} u + b_1 u^2 k_2 - \lambda & a_{12} u + b_1 u^2 k_3 & b_1 u^2 k_1 \frac{1}{x_d} \\ b_2 u^2 k_2 & a_{21} u + b_2 u^2 k_2 & a_{22} u + b_2 u^2 k_3 - \lambda & b_2 u^2 k_1 \frac{1}{x_d} \\ u & 1 & 0 & 0-\lambda \end{vmatrix} = 0 \tag{2.30}$$

III. COMPUTER SIMULATION

A. PROGRAMMING

1. Program in MATRIX.X

The MATRIX.X software is in VAX/VMS in the Mechanical Engineering Department. The program in MATRIX.X is written to find gains and poles of the system from the given inputs, forward speed (u), settling time (t_c) and vehicle lookahead distance (x_d). This is used to compute the eigenvalues of the complete system (Equation (2.30).

The four eigenvalues of the system are computed for the given values of u, t_c and x_d. For stable vehicle response, all four must be negative (or have negative real parts). If at least one is positive, the vehicle response will be unstable and convergence to the straight line path is not to be expected. In such a case, the parameters (in particular the lookahead distance must be changed) so that the vehicle is stable. A listing of this program is presented in Appendix A.

2. Program in FORTRAN

The first program in FORTRAN, is written to find distance along body fixed axis (x-y) from inputs, heading angle (ψ), perpendicular distance from vehicle to route (y), yaw rate (r), forward speed (u), settling time (t_c) and vehicle lookahead distance (x_d). A modified version of this simulation program is presented in Appendix B. The modified version is used to control the vehicle position in a general (X-Y) inertial system. The desired vehicle path is discretized into a series of straight line segments and the same lookahead distance x_d is used to regulate the vehicle deviation of each segment. Details of this modification are presented in the next paragraph.

3. Graphics

The GRAFSTAT graphic package is used to produce 2 dimensional graphs by using data from the simulation programs.

B. DETAILS

1. Poles and gains

Calculate the system poles and gains for a given forward speed u and various combinations of t_c and x_d. Select t_c and x_d such that appropriate (sufficiently negative) poles and gains for the system are produced. This is verified by repeated simulations from step 2 that follows

2. Distance in x-y axis

Using the values of t_c and x_d from the previous step, the system response can be simulated. Unlike the control law design, the simulation is based on the full nonlinear equations of motion for the vehicle, (3), (4), (5) and (7). Typically, the initial conditions consist of nonzero values of the lateral deviation y and heading ψ.

3. Distance in X-Y axis

A similar procedure is used to simulate the vehicle response in a general path in the X-Y plane, as shown in Figure 2. The perpendicular distance y from the vehicle to the desired route in the X-Y plane is used to compute the commanded heading angle

ψ_c. The difference ψ-α, where α is the angle of the route with respect to the X-axis, is used instead of the heading ψ in the control law. The above computations are performed by appropriate coordinate rotations between the two coordinate systems.

When transiting from one straight line path to the next, the same lookahead distance x_d is used for both segments. The vehicle switches to the next segment when it gets within a specified distance from the terminal way point. This distance is measured along the x-axis and for given t_c and x_d, and it should increase as the angle α for the next segment increases. Too high or too low values of this turning distance result in path overshoot and undesirable oscillatory response. The optimum turning distance that allows for the smooth transition between consecutive straight line paths is established with the aid of the simulation program from step 2 as follows.

For a fixed initial heading ψ, the initial deviation y is varied until the vehicle response is smooth and sufficiently fast with no path overshoot. The process is repeated for different initial conditions in ψ and a curve in y versus ψ is constructed. This is shown in Figure 3 and is the desirable turning distance versus turning angle curve. The actual curve is approximated by two straight lines which are used to initiate the turn in the simulation program. A copy of this simulation program is included in Appendix B.



Figure 2 Angles and axes



Figure 3 Turning distance and turning angle

IV. SIMULATION RESULTS

The vehicle parameters used in the simulation are:

$x_{\dot{u}}$	=	-0.0076	$y_{\dot{r}}$	=	0.0012	
x_{rr}	=	0.0040	$y_{\dot{v}}$	=	-0.0550	
x_{vr}	=	0.0200	y_{r}	=	0.0300	
$x_{r\delta}$	=	-0.0010	y_{v}	=	-0.1000	
x_{vv}	=	0.0530	y_{δ}	=	0.0270	
$x_{v\delta}$	=	0.00173	c_{Dy}	=	0.3500	
$x_{\delta\delta}$	=	-0.1000	W	=	12000	*lb.*
$n_{\dot{r}}$	=	-0.0034	l	=	17.4	*ft.*
$n_{\dot{v}}$	=	0.0012	ρ	=	1.94	$slug/ft^3$.
n_{r}	=	-0.0160	g	=	32.2	*ft.*/sec.
n_{v}	=	-0.0074	I_z	=	10000	ft^4.
n_{δ}	=	-0.0130	ν	=	0.000847	ft^2./sec.

The simulation begins by setting $\psi = 5$ degrees, $x_d = 2$ vehicle lengths, $t_c = 5$, $r = 0$, $v = 0$ ft./sec, $u = 5$ ft./sec. and $y = 1$. When the vehicle moves to a distance $x = 20$ then the simulation stops. The route of vehicle and the rudder angle (δ) that vehicle used during simulation are shown in Figure 4. The heading angle (ψ) and commanded heading (ψ_c) are shown in Figure 5. Yaw velocity and sway velocity are shown in Figure 6.

The values for t_c and x_d were selected based on the results of the previous chapter. From the figures it can be seen that the vehicle response is very fast with limited overshoot. This, of course, depends heavily on the initial conditions of the simulation. The actual heading angle converges rather rapidly, after the initial transients have died out, to the commanded heading angle.

The second series of simulations was performed in order to assess the capability of the control and guidance law to change course and keep the new path. Simulation parameters were $x_d = 2$, $t_c = 5$ and $u = 5$ as before. Initial conditions for the simulations were $\psi = 0$, $r = 0$ and $v = 0$ with the initial vehicle position at $(X_0, Y_0) = (5,0)$ in the global reference frame. The first straight line segment is determined by the way points (5,0) and (25,0) and the second by (25,0) and (67.89,20). For this route the corresponding course change is 25 degrees. The results of this simulation are presented in Figure 7 where along with the actual vehicle path, a side path at distance of 1 vehicle length off the desired path is shown. This corresponds to an arbitrary " safety path band " for the vehicle. The turning distance was fixed at 2 vehicle lengths throughout the simulations. From Figure 7 it can be seen that the vehicle turns to the new course smoothly with no path overshoot. When the second route changes to (25,0) and (41.78,20) which corresponds to 50 degrees course change, the results of Figure 8 show that a path overshoot occurs although it is yet not serious enough according to the artificial safety criterion described above. However, when the second route changes to (25,0) and (30.36,20) which corresponds to 75 degrees course change, significant vehicle oscillatory response and side path overshoot occurs, as demonstrated in Figure 9.

The above simulations demonstrate the need for adjustable turning distance; although path accuracy is obtained regardless of the value of the turning distance, the transient response during course change is not always within some predetermined safety bounds. For this reason we employ the built-in turning distance versus turning angle relationship shown in Figure 3 and repeat the simulations for the aforementioned three course changes. The results are shown in Figures 10, 11 and 12, where it can be seen that the vehicle response is now satisfactory for both course keeping and course changing.

Finally, the last simulation test was performed in order to establish the capabilities of the scheme to follow a general path in the horizontal plane. For demonstration purposes the path was assumed to consist of the way points (0,10), (5,10), (25,0), (35,20), (50,20), (70,10), (50,-10) and (30,-10). Straight line segments were assumed as the desired paths between consecutive way points. The same simulation parameters were used as in the previous runs. The results of this simulation are presented in Figures 13 for adjustable turning distance, and 14 for the fixed turning distance case. It can be clearly seen that when the turning distance is function of turning angle, the scheme achieves excellent path keeping characteristics with smooth course changes and minimal path overshoot.



Figure 4 Pursuit navigation



Figure 5 Pursuit navigation



Figure 6 Pursuit navigation



Figure 7 Path control



Figure 8 Path control



Figure 9 Path control



Figure 10 Path control



Figure 11 Path control



Figure 12 Path control



Figure 13 Pure pursuit navigation



Figure 14 Pure pursuit navigation

V. SUMMARY AND CONCLUSIONS

The principal conclusions of this work can be summarized as follows:

1. Orientation control law can be used in order to provide accurate vehicle path keeping when combined with an appropriate guidance scheme.

2. Pure pursuit guidance was found to work very well for the autonomous underwater vehicle case and its simplicity make it a very attractive alternative to cross track error schemes.

3. A built-in turning distance versus course change relationship can be utilized to initiate the turn at the appropriate time in order to avoid path overshoot and achieve smooth path transitions.

4. It is expected that the added robustness that heading schemes naturally enjoy will aid in maintaining stability in cases where incomplete and inaccurate vehicle dynamic descriptions are available.

Some recommendations for further research are as follows:

1. Comparative studies must be performed with other orientation guidance schemes such as proportional navigation and also with velocity guidance laws in order to ensure that the best technique is ultimately employed.

2. Similar studies must be performed in the vertical plane. Combined with the horizontal plane techniques developed in this work and with propulsion control they could be utilized to provide accurate trajectory following in 3-D space.

LIST OF REFERENCES

1. Kanayama, Y. and Hartman, B.I. (1989) " Smooth local path planning for autonomous vehicles, " Proceeding, IEEE International Conference on Robotics and Automation, Scottsdale, Arizona.

2. Lienard, D.L., (1990) " Autopilot design for autonomous underwater vehicles based on sliding mode control, " M.E. Thesis, Mechanical Engineering, Naval Postgraduate School, Monterey California.

3. Chism, S., (1990) " Robust path tracking of autonomous underwater vehicles using sliding modes, " M.E. Thesis, Mechanical Engineering, Naval Postgraduate School, Monterey, California.

4. Hawkinson, T., (1990) " Multiple input sliding mode control for autonomous diving and steering of underwater vehicles, " M.E. Thesis, Mechanical Engineering, Naval Postgraduate School, Monterey, California.

5. Tan, C.H., (1986) " A simulation study of an autonomous steering system for on-road operation of automotive vehicles, " M.S. Thesis, Department of Computer Science, Naval Postgraduate School, Monterey, California.

6. Smith, N.S., Crane, J.W., and Summey, D.C., (1978) " SDV simulator hydrodynamic coefficients, " Naval Coastal Systems Center, Panama City, Florida, Report No. NCSC-TM231-78.

APPENDIX A.

```
l=17.425;
inquire xd_l
inquire tc
inquire u
xd=xd_l*l;
aa11=-0.045380;
aa12=-0.351190;
aa21=-0.002795;
aa22=-0.095680;
bb1= 0.011432;
bb2=-0.004273;
OMEGA=(10.0*U)/(TC*L);
AD1=1.75*OMEGA;
AD2=2.15*OMEGA**2;
AD3=OMEGA**3;
A1=BB1*U*U;
B1=BB2*U*U;
C1=-AD1-(AA11+AA22)*U;
A2=(BB1*AA22-BB2*AA12)*U**3;
B2=(BB2*AA11-BB1*AA21)*U**3;
K1=AD3/((BB2*AA11-BB1*AA21)*U**3);
C2=AD2-(AA11*AA22-AA12*AA21)*U**2+BB2*U*U*K1;
K2=(C1*B2-C2*B1)/(A1*B2-A2*B1);
K3=(C2*A1-C1*A2)/(A1*B2-A2*B1);
a11=0;
a12=0;
a13=1;
a14=0;
a21=bb1*u*u*k1;
a22=aa11*u+bb1*u*u*k2;
a23=aa12*u+bb1*u*u*k3;
a24=bb1*u*u*k1/xd;
a31=bb2*u*u*k1;
a32=aa21*u+bb2*u*u*k2;
a33=aa22*u+bb2*u*u*k3;
a34=bb2*u*u*k1/xd;
a41=u;
a42=1;
a43=0;
a44=0;
a=[a11,a12,a13,a14;a21,a22,a23,a24;
   a31,a32,a33,a34;a41,a42,a43,a44];
eig(a)
```

APPENDIX B.

```
      PROGRAM SUB.FOR
C
C     PROUTTICHAI SUWANDEE
C     NAVAL POSTGRADUATE SCHOOL
C     MARCH 1991
C
C     AUV LINE-OF-SIGHT NAVIGATION AND CONTROL
C     VARIABLE GAINS INTERNALLY COMPUTED
C
      REAL L,MASS,NRDOT,NVDOT,NR,NV,NDR
      REAL IZ,NU,LLL,NSL,K1,K2,K3
      DIMENSION X(9),HH(9),VEC1(9),VEC2(9),TT(1000),YY(6,1000),
     *ALPHA(10),XZ(10),YZ(10)
C
C     LONGITUDINAL HYDRODYNAMIC COEFFICIENTS
C
      PARAMETER(XRR=4.E-3,XUDOT=-7.6E-3,XVR=2.E-2,XRDR=-1.E-3,
     1          XVV=5.3E-2,XVDR=1.7E-3,XDRDR=-1.E-2)
C
C     LATERAL HYDRODYNAMIC COEFFICIENTS
C
      PARAMETER(YRDOT=1.2E-3,YVDOT=-5.5E-2,YR=3.E-2,YV=-1.E-1,
     1          YDR=2.7E-2,CDY=3.5E-1)
C
C     YAW HYDRODYNAMIC COEFFICIENTS
C
      PARAMETER(NRDOT=-3.4E-3,NVDOT=1.2E-3,NR=-1.6E-2,NV=-7.4E-3,
     1          NDR=-1.3E-2)
C
C     MASS CHARACTERISTICS OF THE FLOODED VEHICLE
C
      PARAMETER(WEIGHT=12000.,XG = 0.,IZ=10000.,L=17.4,RHO=1.94,
     1          G=32.2,NU=8.47E-4)
C
      OPEN (10,FILE='SUB.IN',STATUS='OLD')
      OPEN (11,FILE='SUB.OUT',STATUS='OLD')
C
      READ (10,*) TARGET
      READ (10,*) TSIM,DELTA,IPRNT
      READ (10,*) PSI,R
      READ (10,*) U
      READ (10,*) TC,VC
C
C     NUMBER OF POSITION OF ROUTE AND POSITION OF VEHICLE
C
      READ (10,*) N,XO,YO
```

```
C
C     POSITION OF ROUTE IN X-Y AXIS
C
      DO 30 I=1,N
      READ (10,*) XZ(I),YZ(I)
   30 CONTINUE
      TARGET=TARGET*L
      TWOPI =8.0*ATAN(1.0)
      PI    =0.5*TWOPI
      DO 999 M=1,N-1
      PSI1=0
      YTURN=0
      XYTURN=0
      PSI=PSI*PI/180.0
C
C     MOVE ORIGINE OF AXIS TO THE FIRST POINT OF THE ROUTE
C
      XF=XZ(M+1)-XZ(M)
      IF (XF.EQ.0) XF=0.0000001
      YF=YZ(M+1)-YZ(M)
      XO=XO-XZ(M)
      IF (XO.EQ.0) XO=0.0000001
      YO=YO-YZ(M)
C
C     ANGLE OF POSITION OF VEHICLE
C
      ALPHA0=ATAN(ABS(YO/XO))
      IF ((YO.GT.0).AND.(XO.LT.0)) ALPHA0=PI-ALPHA0
      IF ((YO.LE.0).AND.(XO.GT.0)) ALPHA0=2*PI-ALPHA0
      IF ((YO.LE.0).AND.(XO.LT.0)) ALPHA0=PI+ALPHA0
C
C     ANGLE OF ROUTE
C
      ALPHA1=ATAN(ABS(YF/XF))
      IF ((YF.GT.0).AND.(XF.LT.0)) ALPHA1=PI-ALPHA1
      IF ((YF.LE.0).AND.(XF.GT.0)) ALPHA1=2*PI-ALPHA1
      IF ((YF.LE.0).AND.(XF.LT.0)) ALPHA1=PI+ALPHA1
C
C     ANGLE BETWEEN VEHICLE AND ROUTE
C
      BETA=(ALPHA0-ALPHA1)
C
C     DISTANCE FROM ORIGINE TO VEHICLE
C
      RT=(XO**2+YO**2)**0.5
C
C     PROJECTED DISTANCE FROM ORIGINE TO VEHICLE ON THE ROUTE
C
      XS=RT*COS(BETA)
      XC=(XS)*COS(ALPHA1)
      YC=(XS)*SIN(ALPHA1)
```

```
C
C     PERPENDICULAR DISTANCE OF VEHICLE TO X-AXIS
C
      YPOS=RT*SIN(BETA)
C
C     TOTAL DISTANCE ON THE ROUTE
C
      TXD=(XF**2+YF**2)**0.5
      XD=ABS(TXD-XS)
C
C     HEADING ANGLE
C
      IF(M.EQ.1) PSI=PSI-ALPHA1
      IF (PSI.GT.PI) PSI=PSI-2*PI
      IF (PSI.LT.-PI) PSI=PSI+2*PI
      IF (M.EQ.N-1) GO TO 65
C
C     NEXT HEADING ANGLE
C
      DUMY=YZ(M+2)-YZ(M+1)
      DUMX=XZ(M+2)-XZ(M+1)
      IF (DUMX.EQ.0) DUMX=0.0000001
      ALP2=ATAN(ABS(DUMY/DUMX))
      IF ((DUMY.GT.0).AND.(DUMX.LT.0)) ALP2=PI-ALP2
      IF ((DUMY.LE.0).AND.(DUMX.GT.0)) ALP2=2*PI-ALP2
      IF ((DUMY.LE.0).AND.(DUMX.LT.0)) ALP2=PI+ALP2
      PSI1=ALPHA1-ALP2
      IF (PSI1.GT.PI) PSI1=PSI1-2*PI
      IF (PSI1.LT.-PI) PSI1=PSI1+2*PI
      PSI1=PSI1*180/PI
C
C     TURNING DISTANCE
C
      IF(ABS(PSI1).LE.45) YTURN=ABS(PSI1/25)
      IF(ABS(PSI1).GT.45) YTURN=(ABS(PSI1)-45)*0.3/5+1.8
      XYTURN=YTURN/ABS(SIN(PSI1*PI/180))-0.2
      IF(PSI1.EQ.0) XYTURN=0.0
C
 65   UC    =U
      OMEGA=(10.0*U)/(TC*L)
      AD1   =1.75*OMEGA
      AD2   =2.15*OMEGA**2
      AD3   =OMEGA**3
C
      PISIM  =TSIM/DELTA
      ISIM   =PISIM
      ECHO   =1.0/DELTA
      IECHO  =IPRNT*20
      YAW    =0.0
      SWAY   =0.0
      V      =0.0
      DR     =0.0
      R      =R*PI/180.0
```

```
      ISTART=1
      XPOS  =0.0
      YPOS  =YPOS*L
C
C     DEFINE THE LENGTH X AND HEIGHT HH TERMS FOR THE INTEGRATION
C
      X(1)  = -105.9/12.
      X(2)  =  -99.3/12.
      X(3)  =  -87.3/12.
      X(4)  =  -66.3/12.
      X(5)  =   72.7/12.
      X(6)  =   83.2/12.
      X(7)  =   91.2/12.
      X(8)  =   99.2/12.
      X(9)  =  103.2/12.
C
      HH(1) =   0.00/12.
      HH(2) =   8.24/12.
      HH(3) =  19.76/12.
      HH(4) =  29.36/12.
      HH(5) =  31.85/12.
      HH(6) =  27.84/12.
      HH(7) =  21.44/12.
      HH(8) =  12.00/12.
      HH(9) =   0.00/12.
C
      MASS = WEIGHT/G
C
      P1=MASS-0.5*RHO*L**3*XUDOT
      P3=MASS-0.5*RHO*L**3*YVDOT
      P4=MASS*XG-0.5*RHO*L**4*YRDOT
      P5=IZ-0.5*RHO*L**5*NRDOT
      P6=MASS*XG-0.5*RHO*L**4*NVDOT
      D =P5*P3-P4*P6
C
      AA11=(P5*0.5*RHO*L*L*YV-P4*0.5*RHO*L**3*NV)/D
      AA12=(P5*(-MASS+0.5*RHO*L**3*YR)-P4*(-MASS*XG
     1     +0.5*RHO*L**4*NR))/D
      AA21=(P3*0.5*RHO*L**3*NV-P6*0.5*RHO*L*L*YV)/D
      AA22=(P3*(-MASS*XG+0.5*RHO*L**4*NR)-P6*(-MASS
     1     +0.5*RHO*L**3*YR))/D
      BB1=(P5*0.5*RHO*L**2*YDR-P4*0.5*RHO*L**3*NDR)/D
      BB2=(P3*0.5*RHO*L**3*NDR-P6*0.5*RHO*L**2*YDR)/D
C
      A1 =BB1*U*U
      B1 =BB2*U*U
      C1 =-AD1-(AA11+AA22)*U
      A2 =(-AA12*BB2+AA22*BB1)*U**3
      B2 =(-AA21*BB1+AA11*BB2)*U**3
      K1=AD3/((BB2*AA11-BB1*AA21)*U**3)
      C2 =AD2-(-AA12*AA21+AA11*AA22)*U**2+BB2*U*U*K1
      K2=(C1*B2-C2*B1)/(A1*B2-A2*B1)
      K3=(C2*A1-C1*A2)/(A1*B2-A2*B1)
```

```
      J =0
      IJ=0
      JE=0
      OFF=0
C
      DRHAT=0.0
      DRBAR=0.0
C
C     SIMULATION BEGINS
C
      DO 100 I=1,ISIM
C
C       CALCULATE THE DRAG FORCE, INTEGRATE THE DRAG OVER THE VEHICLE
C
        DO 600 K=1,9
          UCF=V+X(K)*R
          SGN=1.0
          IF (UCF.LT.0.0) SGN=-1.0
          VEC1(K)=HH(K)*UCF*UCF*SGN
          VEC2(K)=HH(K)*UCF*UCF*SGN*X(K)
  600   CONTINUE
        CALL TRAP(9,VEC1,X,SWAY)
        CALL TRAP(9,VEC2,X,YAW)
        SWAY=-0.5*RHO*CDY*SWAY
        YAW =-0.5*RHO*CDY*YAW
C
C       FORCE EQUATIONS
C
        FP2 = -MASS*U*R+0.5*RHO*L**3*YR*U*R+0.5*RHO*L*L*(
     1          YV*U*V+YDR*U*U*DR)+SWAY
C
        FP3 = -MASS*XG*U*R+0.5*RHO*L**4*NR*U*R+0.5*RHO*L**3*
     1          (NV*U*V+NDR*U*U*DR)+YAW
C
        VDOT  =(P5*FP2-P4*FP3)/(P5*P3-P4*P6)
        RDOT  =(P6*FP2-P3*FP3)/(P4*P6-P3*P5)
        PSIDOT=R
        YDOT  =U*SIN(PSI)+V*COS(PSI)+VC
        XDOT  =U*COS(PSI)-V*SIN(PSI)
C
C       FIRST ORDER INTEGRATION
C
        PSI =PSI  +DELTA*PSIDOT
        V   =V    +DELTA*VDOT
        R   =R    +DELTA*RDOT
        XPOS=XPOS+DELTA*XDOT
        YPOS=YPOS+DELTA*YDOT
C
        YCTE=YPOS
        XAWAY=(XPOS-XD*L)
C
        IF ((XAWAY).GE.-(XYTURN*L)) OFF=1
```

```
C
C       RUDDER INPUT CALCULATION
C
        YA=ABS(YPOS)
        HDM=ATAN((YPOS)/(-TARGET))
        DR=K1*(PSI-HDM)+K2*V+K3*R
C
        IF (DR.GT. 0.4) DR= 0.4
        IF (DR.LE.-0.4) DR=-0.4
C
C       PRINT RESULTS
C
        JE=JE+1
        IF (JE.NE.IECHO) GO TO 99
        WRITE (*,*) ' XAWAY =', XAWAY/L
        JE=0
   99   J=J+1
        IF (J.NE.IPRNT) GO TO 100
        IJ=IJ+1
        TIME=I*DELTA
        XP=XPOS/L
        YP=YPOS/L
        XI=XZ(M)+XC+XP*COS(-ALPHA1)+YP*SIN(-ALPHA1)
        YI=YZ(M)+YC+YP*COS(-ALPHA1)-XP*SIN(-ALPHA1)
        WRITE (11,*) XI,YI
        J=0
        IF (OFF.EQ.1) GO TO 500
  100   CONTINUE
  500   PSI=PSI1
        XO=XI
        YO=YI
  999   CONTINUE
        STOP
        END
C
      SUBROUTINE TRAP(N,A,B,OUT)
C
C     NUMERICAL INTEGRATION ROUTINE USING THE TRAPEZOIDAL RULE
C
      DIMENSION A(1),B(1)
      N1=N-1
      OUT=0.0
      DO 1 I=1,N1
        OUT1=0.5*(A(I)+A(I+1))*(B(I+1)-B(I))
        OUT =OUT+OUT1
    1 CONTINUE
      RETURN
      END
```

INITIAL DISTRIBUTION LIST

1.	Defense Technical Information Center Cameron Station Alexandria, Virginia 22304-6145	2
2.	Library, Code 52 Naval Postgraduate School Monterey, California 93943-5002	2
3.	Superintendent, Code 34 Naval Postgraduate School Monterey, California 93943	1
4.	Prof. F.A. Papoulias, Code ME/PA Naval Postgraduate School Monterey, California 93943	3
5.	Chairman, Dept. of Mechanical Engineering Naval Postgraduate School Monterey, California 93943	1
6.	LCDR. Prouttichai Suwandee 301/814, Sukhapiban 3 Rd Bangkapi, Bangkok 10240 Thailand	2
7.	LCDR. Damp Pinchaliaw 67/63, Ladprao 113 Bangkapi, Bangkok 10240 Thailand	1